
[!NOTE]
Peri is implementing an Ecto integration right now, you can test it pointing your mix.exs def to the feat/ecto-integration branch
Something like {:peri, github: "zoedsoupe/peri", branch: "feat/ecto-integration"}
You can follow the development status on https://github.com/zoedsoupe/peri/pull/18

Peri is a schema validation library for Elixir, inspired by Clojure's Plumatic Schema. It provides a powerful and flexible way to define and validate schemas for your data, ensuring data integrity and consistency throughout your application. Peri supports a variety of types and validation rules, and it can generate sample data based on your schemas.

  
    
  
  Features


	Schema Definition: Define schemas using a concise and expressive DSL.
	Nested Schemas: Support for deeply nested and complex schemas.
	Custom Validation: Implement custom validation functions for specific requirements.
	Error Handling: Detailed error messages with path information for easy debugging.
	Data Generation: Generate sample data based on your schemas using StreamData.


  
    
  
  Installation


Add this line to your mix.exs:
defp deps do
  [
    {:peri, "~> 0.3"}
  ]
end

  
    
  
  Available Types


	:any - Allows any data type.
	:atom - Validates that the field is an atom.
	:string - Validates that the field is a binary (string).	{:regex, regex} - Validates that the string field matches a given regex
	{:eq, val} - Validates that the string field is equal to val
	{:min, min} - Validates that the string field has at least the min length
	{:max, max} - Validates that the string field has at maximum the max length


	:integer - Validates that the field is an integer.	{:eq, val} - Validates taht the integer field is equal to val
	{:neq, val} - Validates taht the integer field is not equal to val
	{:lt, val} - Validates taht the integer field is lesss than val
	{:lte, val} - Validates taht the integer field is less than or equal to val
	{:gt, val} - Validates taht the integer field is greater than val
	{:gte, val} - Validates taht the integer field is greater than or equal to val
	{:range, {min, max}} - Validates taht the integer field is inside the range of min to max (inclusive)


	:float - Validates that the field is a float.
	:boolean - Validates that the field is a boolean.
	:map - Validates that the field is a map.
	{:map, type} - Validates that the field is a map where all values conform to the specified type.
	{:map, key_type, value_type} - Validates that the field is a map where keys conform to key_type and values conform to value_type.
	{:required, type} - Marks the field as required and validates it according to the specified type.
	{:enum, choices} - Validates that the field is one of the specified choices.
	{:list, type} - Validates that the field is a list of elements of the specified type.
	{:literal, value} - Validates that the field is exactly equal to the specified literal value.
	{:tuple, types} - Validates that the field is a tuple with elements of the specified types.
	{type, {:default, default}} - Provides a default value if the field is missing or nil.	{type, {:default, &some_fun/0}} - The default values is retrieved from callinf some_fun/0 if the field is missing.
	{type, {:default, {mod, fun}}} - The default values is retrieved from callinf mod.fun/0 if the field is missing.


	{type, {:transform, mapper}} - Transforms the field value using the specified mapper function. It can be a 1 or 2 arity function: when is a single arity the mapper function will only receive the defined field value, while with 2 arity will receive the current defined field value and the whole data as the second argument.	{type, {:transform, {mod, fun}}} - Transforms the field value using the specified mod.fun/1 function. Notice that fun can be a 2 arity so it can receive the whole data being validated, in case on dependent fields transformations.
	{type, {:transform, {mod, fun, args}}} - Transforms the field value using the specified MFA. Notice that fun will be at least a 2 arity one so it can receive the whole data being validated, in case on dependent fields transformations and the maximum arity allowed will be 2 + length(args).


	{:either, {type1, type2}} - Validates that the field is either of the two specified types.
	{:oneof, types} - Validates that the field is one of the specified types.
	{:custom, callback} - Validates that the field passes the custom validation function.
	{:custom, {mod, fun}} - Validates that the field passes the custom validation function.
	{:custom, {mod, fun, args}} - Validates that the field passes the custom validation function.
	{:dependent, field, condition, type} - Validates the field based on the value of another field. Check the dependent schema examples section for more info.
	{:dependent, condition} - Validates the field based on the value of multiple data values. Check the dependent schema examples section for more info.	{:dependent, {mod, fun}} - Validates the field based on the value of multiple data values but executes mod.fun/1 in runtime.


	{:cond, condition, type, else_type} - Conditional validation based on a condition function. Check the conditional schema examples section for more info.


  
    
  
  Defining Schemas



  
    
  
  Using the Macro


You can define schemas using the defschema macro, which provides a concise syntax for defining and validating schemas.
defmodule MySchemas do
  import Peri

  defschema :user, %{
    name: :string,
    age: {:integer, {:transform, & &1 * 2}},
    email: {:required, :string},
    address: %{
      street: :string,
      city: :string
    },
    tags: {:list, :string},
    role: {:enum, [:admin, :user, :guest]},
    status: {:literal, :active},
    preferences: {:map, :string},
    scores: {:map, :string, :integer},
    geolocation: {:tuple, [:float, :float]},
    rating: {:custom, &validate_rating/1}
  }

  defp validate_rating(n) when n < 10, do: :ok
  defp validate_rating(_), do: {:error, "invalid rating", []}
end

  
    
  
  Without the Macro


You can also define schemas directly without using the macro:
schema = %{
  name: :string,
  age: {:integer, {:transform, & &1 * 2}},
  email: {:required, :string}
}

Peri.validate(schema, %{name: "John", age: 30, email: "john@example.com"})

  
    
  
  Composable and Reusable Schemas


Schemas can be composed and reused to build complex data structures.
defmodule MySchemas do
  import Peri

  defschema :address, %{
    street: :string,
    city: :string
  }

  defschema :user, %{
    name: :string,
    age: :integer,
    email: {:required, :string},
    address: get_schema(:address)
  }
end

  
    
  
  Conditional Schemas


You can define conditional types for a schema based on a callback condition, let's see an example:
defmodule CondSchema do
  import Peri

  defschema(:details, %{
    email: {:required, :string},
    country: {:required, :string}
  })

  defschema(:info, %{
    name: {:required, :string},
    provide_details: {:required, :boolean},
    details: {:cond, & &1.provide_details, {:required, get_schema(:details)}, nil}
  })
end
In this example we can read the info.details field schema definition as: "if the provide_details field is true then the info.details field should be parsed as the details schema, else, it should be parsed as nil".
Notice that the condition callback should return boolean


  
    
  
  Dependent Schemas


You can parse fields that depend on onther fields, let's check some examples
Single field dependency
defmodule UserSchemas do
  import Peri

  defschema :user_registration, %{
    username: {:required, :string},
    password: {:required, :string},
    password_confirmation: {:dependent, :password, &validate_confirmation/2, :string}
  }

  # if confirmation has the same value of password, the validation is ok
  defp validate_confirmation(%{password: password}, password), do: :ok

  defp validate_confirmation(_confirmation, _password) do
    {:error, "confirmation should be equal to password", []}
  end
end
In this example we say that "the user_registration.password_confirmation field should be parsed as string only if it passes the validate_confirmation/2 function, which in this case asserts that the user_registration.password field should be equal to the confirmation one."
The callback passed to this type definition should be a 2 arity function that will receive the current nest level data as first argument and the value of the current field as the second argument and it should return only :ok or {:error, template, context}.
Multiple fields dependencies and custom parsing
A more complex dependent type schema definition would be:
defmodule TypeDependentSchema do
  import Peri

  defschema(:email_details, %{email: {:required, :string}})

  defschema(:country_details, %{country: {:required, :string}})

  defschema(:details, Map.merge(get_schema(:email_details), get_schema(:country_details)))

  defschema(:info, %{
    name: {:required, :string},
    provide_email: {:required, :boolean},
    provide_country: {:required, :boolean},
    details: {:dependent, &verify_details/1}
  })

  defp verify_details(%{data: data}) do
    %{provide_email: pe, provide_country: pc} = data

    provide = {pe, pc}

    case provide do
      {true, true} -> {:ok, {:required, get_schema(:details)}}
      {true, false} -> {:ok, {:required, get_schema(:email_details)}}
      {false, true} -> {:ok, {:required, get_schema(:country_details)}}
      {false, false} -> {:ok, nil}
    end
  end
end
In this example we have different schemas parsing rules based on the structure and values of the given data. Basically this type deifinition could be read as:
	if the field info.provide_email and info.provide_country is both true, then the info.details field is required to provide both email and country fields.
	if the field info.provide_email is true but info.provide_country is false, so info.details should only contains the info.details.email field.
	if the field info.provide_email is false but info.provide_country is true, so info.details should only contains the info.details.country field.
	if both info.provide_email and info.provide_country are false, so info.details should be parsed as nil.

Notice that this kind of dependent type definition should return {:ok, type} whereas type is a valid Peri schema, or {:error, template, context}.

  
    
  
  Custom Validation Functions


Implement custom validation functions to handle specific validation logic.
The spec of the custom validation function is:
@spec validation(term) :: :ok | {:error, template :: String.t(), context :: map | keyword}
Where template is a template string with the notation of %{value} where value is the name of the variable to be injected on the template. And context is a map or keyword list where the key is the name of the variable that will be injected into the template and the value is the value of this injected variable. Let's see an example:
defmodule MySchemas do
  import Peri

  defschema :user, %{
    name: :string,
    age: {:custom, &validate_age/1}
  }

  defp validate_age(age) when age >= 0 and age <= 120, do: :ok
  defp validate_age(age), do: {:error, "invalid age, received: %{age}", [age: age]}
end

  
    
  
  Error Handling with Peri.Error


Peri provides detailed error messages to help identify validation issues. Errors include path information to pinpoint the exact location of the error in the data structure.
case Peri.validate(schema, data) do
  {:ok, valid_data} -> IO.puts("Data is valid!")
  {:error, errors} -> IO.inspect(errors, label: "Validation errors")
end

  
    
  
  Data Generation


Peri can generate sample data based on your schemas using StreamData.
For this feature to work, ensures that you application depends on stream_data.
schema = %{
  name: :string,
  age: {:integer, {:gte, 18}},
  active: :boolean
}

sample_data = Peri.generate(schema)
Enum.take(sample_data, 10) # Generates 10 samples of the schema

  
    
  
  Perfect for Raw Data Structures


Peri excels in validating raw data structures, such as tuples, strings, lists, and integers, with extensive validation options. This makes it ideal for use cases where you need to enforce strict data integrity rules on a wide variety of data types. Here's how Peri can help you handle these data structures:

  
    
  
  Tuples


Tuples can be validated for their structure and content, ensuring each element meets specific criteria.
defmodule MySchemas do
  import Peri

  defschema :coordinates, {:tuple, [:float, :float]}
end

data = {12.34, 56.78}
Peri.validate(get_schema(:coordinates), data)
# => {:ok, {12.34, 56.78}}

invalid_data = {12.34, "not a float"}
Peri.validate(get_schema(:coordinates), invalid_data)
# => {:error, [%Peri.Error{message: "expected type of :float received \"not a float\" value"}]}

  
    
  
  Strings


Strings can be validated for length, equality, and matching regular expressions.
defmodule MySchemas do
  import Peri

  defschema :username, {:string, {:regex, ~r/^[a-zA-Z0-9_]+$/}}
end

valid_data = %{username: "valid_user"}
Peri.validate(get_schema(:username), valid_data)
# => {:ok, %{username: "valid_user"}}

invalid_data = %{username: "invalid user"}
Peri.validate(get_schema(:username), invalid_data)
# => {:error, [%Peri.Error{message: "should match the ~r/^[a-zA-Z0-9_]+$/ pattern"}]}

  
    
  
  Lists


Lists can be validated to ensure all elements are of a specific type and meet certain criteria.
defmodule MySchemas do
  import Peri

  defschema :tags, {:list, :string}
end

valid_data = %{tags: ["elixir", "programming"]}
Peri.validate(get_schema(:tags), valid_data)
# => {:ok, %{tags: ["elixir", "programming"]}}

invalid_data = %{tags: ["elixir", 42]}
Peri.validate(get_schema(:tags), invalid_data)
# => {:error, [%Peri.Error{message: "expected type of :string received 42 value"}]}

  
    
  
  Integers


Integers can be validated for equality, inequality, and range constraints.
defmodule MySchemas do
  import Peri

  defschema :age, {:integer, {:range, {18, 65}}}
end

valid_data = %{age: 30}
Peri.validate(get_schema(:age), valid_data)
# => {:ok, %{age: 30}}

invalid_data = %{age: 17}
Peri.validate(get_schema(:age), invalid_data)
# => {:error, [%Peri.Error{message: "should be in the range of 18..65 (inclusive)"}]}

  
    
  
  Maps with Type Constraints


Maps can be validated to ensure all values match a specific type or that both keys and values match specified types.
defmodule MySchemas do
  import Peri

  defschema :user_preferences, {:map, :string}
  defschema :user_scores, {:map, :string, :integer}
end

# Map with string values
valid_prefs = %{theme: "dark", notifications: "enabled"}
Peri.validate(get_schema(:user_preferences), valid_prefs)
# => {:ok, %{theme: "dark", notifications: "enabled"}}

# Map with string keys and integer values
valid_scores = %{"math" => 95, "science" => 92}
Peri.validate(get_schema(:user_scores), valid_scores)
# => {:ok, %{"math" => 95, "science" => 92}}

# Invalid data with wrong value type
invalid_prefs = %{theme: "dark", notifications: true}
Peri.validate(get_schema(:user_preferences), invalid_prefs)
# => {:error, [%Peri.Error{message: "expected type of :string received true value"}]}

  
    
  
  Literal Values


Literal values can be validated to ensure they exactly match a specific value.
defmodule MySchemas do
  import Peri

  defschema :user_status, {:literal, :active}
end

valid_data = :active
Peri.validate(get_schema(:user_status), valid_data)
# => {:ok, :active}

invalid_data = :inactive
Peri.validate(get_schema(:user_status), invalid_data)
# => {:error, [%Peri.Error{message: "expected literal value :active but got :inactive"}]}

  
    
  
  Comprehensive Validation Options


Peri's robust validation capabilities make it suitable for various data types and validation needs:
	Equality and Inequality: Validate that values match or do not match specific criteria.
	Ranges: Ensure numerical values fall within specified bounds.
	Regular Expressions: Enforce patterns on string data.
	Map Constraints: Validate maps with type constraints on keys and values.
	Literal Values: Ensure exact matches with specified values. 
	Custom Validation: Implement complex rules through custom functions.

By supporting these raw data structures and providing detailed error handling, Peri ensures that your data remains consistent and adheres to the defined rules, making it an excellent choice for applications requiring strict data validation.

  
    
  
  Comparison with other data validation and mapping libraries



  
    
  
  Peri vs. Norm


Norm is another Elixir library for schema and data validation. While it shares some similarities with Peri, there are distinct differences:
	Focus: Norm focuses on defining specifications and generating tests, offering a more comprehensive approach to data specifications, while Peri focuses on schema validation and transformations.
	Validation: Peri provides a more extensive set of built-in validations for strings, numbers, and custom types, whereas Norm allows for more generic and composable specifications.


  
    
  
  Peri vs. Drops


Drops is another Elixir library designed for validating and casting data. Key differences include:
	Schema Definition: Drops uses a different syntax and approach for defining schemas. Peri provides a more familiar and flexible schema definition approach, not relying on macros.
	Validation Capabilities: Peri offers more advanced validation features, such as default values, transformations, and custom validations.
	Error Handling: Both libraries provide detailed error handling, but Peri's Peri.Error struct offers more context and customization options for error messages.


  
    
  
  Peri vs. Ecto Schemaless Changesets


Ecto is a powerful data mapping and query generator for Elixir, and it offers schemaless changesets for validating data without defining database schemas.
	Flexibility: Peri offers more flexibility in defining schemas for various data structures, such as tuples, nested maps, and lists, along with detailed validation options.
	Composable Validations: Peri allows for more composable and reusable schemas, making it ideal for scenarios where data structures are complex and nested.


  
    
  
  Peri vs. Ecto Embedded Changesets


Ecto embedded changesets are used for validating and casting nested structures within Ecto schemas.
	Nested Structures: Both Peri and Ecto handle nested structures well. However, Peri provides more granular control over validation rules for different types of nested data.
	Usage Context: Ecto embedded changesets are tightly integrated with Ecto schemas and structs, whereas Peri can be used independently of any kind of data structure or type, making it more versatile for various use cases.
	Custom Validations: Peri's custom validation functions and integration with StreamData for data generation provide additional capabilities not available in Ecto.


  
    
  
  Summary


While all these libraries offer data validation capabilities, Peri stands out with its flexibility, comprehensive validation options, and integration with StreamData for data generation. Whether you're dealing with raw data structures, need advanced validation features, or want to generate test data, Peri provides a robust and versatile solution tailored to meet these needs.

  
    
  
  Why the Name "Peri"?


The name "Peri" is derived from the Greek word "περί" (pronounced "peri"), which means "around" or "about." This name was chosen to reflect the library's primary purpose: to provide comprehensive and flexible schema validation for data structures in Elixir. Just as "peri" suggests encompassing or surrounding something, Peri aims to cover all aspects of data validation, ensuring that data conforms to specified rules and constraints.
The choice of the name "Peri" also hints at the library's ability to handle a wide variety of data types and structures, much like how the term "around" can denote versatility and inclusiveness. Whether it's validating nested maps, complex tuples, or strings with specific patterns, Peri is designed to be a robust tool that can adapt to various validation needs in Elixir programming.

Peri is a schema validation library for Elixir, inspired by Clojure's Plumatic Schema.
It provides a flexible and powerful way to define and validate data structures using schemas.
The library supports nested schemas, optional fields, custom validation functions, and various type constraints.

  
    
  
  Key Features


	Simple and Nested Schemas: Define schemas that can handle complex, nested data structures.
	Optional and Required Fields: Specify fields as optional or required with type constraints.
	Custom Validation Functions: Use custom functions to validate fields.
	Comprehensive Error Handling: Provides detailed error messages for validation failures.
	Type Constraints: Supports various types including enums, lists, maps, tuples, literals, and more.


  
    
  
  Usage


To define a schema, use the defschema macro. By default, all fields in the schema are optional unless specified otherwise.
defmodule MySchemas do
  import Peri

  defschema :user, %{
    name: :string,
    age: :integer,
    email: {:required, :string},
    address: %{
      street: :string,
      city: :string
    },
    tags: {:list, :string},
    role: {:enum, [:admin, :user, :guest]},
    geolocation: {:tuple, [:float, :float]},
    preferences: {:map, :string},
    scores: {:map, :string, :integer},
    status: {:literal, :active},
    rating: {:custom, &validate_rating/1}
  }

  defp validate_rating(n) when n < 10, do: :ok
  defp validate_rating(_), do: {:error, "invalid rating", []}
end
You can then use the schema to validate data:
user_data = %{
  name: "John", age: 30, email: "john@example.com",
  address: %{street: "123 Main St", city: "Somewhere"},
  tags: ["science", "funky"], role: :admin,
  geolocation: {12.2, 34.2}, 
  preferences: %{"theme" => "dark", "notifications" => "enabled"},
  scores: %{"math" => 95, "science" => 92},
  status: :active,
  rating: 9
}

case MySchemas.user(user_data) do
  {:ok, valid_data} -> IO.puts("Data is valid!")
  {:error, errors} -> IO.inspect(errors, label: "Validation errors")
end

  
    
  
  Error Handling


Peri provides detailed error messages that include the path to the invalid data, the expected and actual values, and custom error messages for custom validations.

  
    
  
  Schema Types


Peri supports the following schema types:
	:string, :integer, :float, :boolean, :atom, :map, :pid - Basic types
	{:required, type} - Mark a field as required
	{:list, type} - List of elements of the given type
	{:map, type} - Map with values of the given type
	{:map, key_type, value_type} - Map with keys and values of specified types
	{:tuple, [type1, type2, ...]} - Tuple with elements of specified types
	{:enum, [value1, value2, ...]} - One of the specified values
	{:literal, value} - Exactly matches the specified value
	{:either, {type1, type2}} - Either type1 or type2
	{:oneof, [type1, type2, ...]} - One of the specified types
	Nested maps for complex structures


  
    
  
  Functions


	validate/2 - Validates data against a schema.
	conforms?/2 - Checks if data conforms to a schema.
	validate_schema/1 - Validates the schema definition.
	generate/1 - Generates sample data based on schema (when StreamData is available).


  
    
  
  Example


defmodule MySchemas do
  import Peri

  defschema :user, %{
    name: :string,
    age: :integer,
    email: {:required, :string}
  }
end

user_data = %{name: "John", age: 30, email: "john@example.com"}
case MySchemas.user(user_data) do
  {:ok, valid_data} -> IO.puts("Data is valid!")
  {:error, errors} -> IO.inspect(errors, label: "Validation errors")
end

        Helper function to put a value into an enum, handling

          @type cond_def() ::
  {:cond, condition :: (term() -> boolean()), true_branch :: schema_def(),
   else_branch :: schema_def()}

          @type custom_def() ::
  {:custom, (term() -> term())}
  | {:custom, {module(), atom()}}
  | {:custom, {module(), atom(), [term()]}}

          @type default_def() ::
  {schema_def(), {:default, term()}}
  | {schema_def(), {:default, (-> term())}}
  | {schema_def(), {:default, {module(), atom()}}}

          @type dependent_def() ::
  {:dependent, field :: atom(), validation(), type :: schema_def()}
  | {:dependent,
     (term() ->
        {:ok, schema_def() | nil}
        | {:error, template :: String.t(), context :: map() | keyword()})}

  :float
  | {:float,
     {:eq, float()}
     | {:neq, float()}
     | {:lt, float()}
     | {:lte, float()}
     | {:gt, float()}
     | {:gte, :float}
     | {:range, {min :: float(), max :: float()}}}

          @type int_def() ::
  :integer
  | {:integer,
     {:eq, integer()}
     | {:neq, integer()}
     | {:lt, integer()}
     | {:lte, integer()}
     | {:gt, integer()}
     | {:gte, integer()}
     | {:range, {min :: integer(), max :: integer()}}}


      



  



  
    
      
    
    
      literal()



        
          
        

    

  


  

      

          @type literal() :: integer() | float() | atom() | String.t() | boolean()

          @type schema() ::
  schema_def()
  | %{required(String.t()) => schema_def()}
  | %{required(atom()) => schema_def()}
  | [{atom(), schema_def()}]

          @type schema_def() ::
  :any
  | :atom
  | :boolean
  | :map
  | :pid
  | {:either, {schema_def(), schema_def()}}
  | {:oneof, [schema_def()]}
  | {:required, schema_def()}
  | {:enum, [term()]}
  | {:list, schema_def()}
  | {:map, schema_def()}
  | {:map, schema_def(), schema_def()}
  | {:tuple, [schema_def()]}
  | {:literal, literal()}
  | time_def()
  | string_def()
  | int_def()
  | float_def()
  | default_def()
  | transform_def()
  | custom_def()


      



  



  
    
      
    
    
      string_def()



        
          
        

    

  


  

      

          @type string_def() ::
  :string
  | {:string,
     {:regex, Regex.t()}
     | {:eq, String.t()}
     | {:min, integer()}
     | {:max, integer()}}


      



  



  
    
      
    
    
      time_def()



        
          
        

    

  


  

      

          @type time_def() :: :time | :date | :datetime | :naive_datetime


      



  



  
    
      
    
    
      transform_def()



        
          
        

    

  


  

      

          @type transform_def() ::
  {schema_def(), {:transform, (term() -> term()) | (term(), term() -> term())}}
  | {schema_def(), {:transform, {module(), atom()}}}
  | {schema_def(), {:transform, {module(), atom(), [term()]}}}


      



  



  
    
      
    
    
      validation()



        
          
        

    

  


  

      

          @type validation() :: (term() ->
                   :ok
                   | {:error, template :: String.t(),
                      context :: map() | keyword()})


      



  


        

      

      
        Functions

        


  
    
      
    
    
      conforms?(schema, data)



        
          
        

    

  


  

Checks if the given data conforms to the specified schema.

  
    
  
  Parameters


	schema: The schema definition to validate against.
	data: The data to be validated.


  
    
  
  Returns


	true if the data conforms to the schema.
	false if the data does not conform to the schema.


  
    
  
  Examples


iex> schema = %{name: :string, age: :integer}
iex> data = %{name: "Alice", age: 30}
iex> Peri.conforms?(schema, data)
true

iex> invalid_data = %{name: "Alice", age: "thirty"}
iex> Peri.conforms?(schema, invalid_data)
false

  



  
    
      
    
    
      defschema(name, schema)


        (macro)


        
          
        

    

  


  

Defines a schema with a given name and schema definition.

  
    
  
  Examples


defmodule MySchemas do
  import Peri

  defschema :user, %{
    name: :string,
    age: :integer,
    email: {:required, :string}
  }
end

user_data = %{name: "John", age: 30, email: "john@example.com"}
MySchemas.user(user_data)
# => {:ok, %{name: "John", age: 30, email: "john@example.com"}}

invalid_data = %{name: "John", age: 30}
MySchemas.user(invalid_data)
# => {:error, [email: "is required"]}

  



  
    
      
    
    
      generate(schema)



        
          
        

    

  


  

Generates sample data based on the given schema definition using StreamData.
This function validates the schema first, and if the schema is valid, it uses the
Peri.Generatable.gen/1 function to generate data according to the schema.
Note that this function returns a Stream, so you traverse easily the data generations.

  
    
  
  Parameters


	schema: The schema definition to generate data for.


  
    
  
  Returns


	{:ok, stream} if the data is successfully generated.
	{:error, errors} if there are validation errors in the schema.


  
    
  
  Examples


iex> schema = %{name: :string, age: {:integer, {:range, {18, 65}}}}
iex> {:ok, stream} = Peri.generate(schema)
iex> [data] = Enum.take(stream, 1)
iex> is_map(data)
true
iex> data[:age] in 18..65
true

  



  
    
      
    
    
      get_enumerable_value(enum, key)



        
          
        

    

  


  


  



  
    
      
    
    
      is_enumerable(data)


        (macro)


        
          
        

    

  


  

Checks if the given data is an enumerable, specifically a map or a list.

  
    
  
  Parameters


	data: The data to check.


  
    
  
  Examples


iex> is_enumerable(%{})
true

iex> is_enumerable([])
true

iex> is_enumerable(123)
false

iex> is_enumerable("string")
false

  



  
    
      
    
    
      is_numeric(n)


        (macro)


        
          
        

    

  


  

Checks if the given data is a numeric value, specifically a integer or a float.

  
    
  
  Parameters


	data: The data to check.


  
    
  
  Examples


iex> is_numeric(123)
true

iex> is_numeric(0xFF)
true

iex> is_numeric(12.12)
true

iex> is_numeric("string")
false

iex> is_numeric(%{})
false

  



  
    
      
    
    
      is_numeric_type(t)


        (macro)


        
          
        

    

  


  

Checks if the given type as an atom is a numeric (integer or float).

  
    
  
  Parameters


	data: The data to check.


  
    
  
  Examples


iex> is_numeric(:integer)
true

iex> is_numeric(:float)
true

iex> is_numeric(:list)
false

iex> is_numeric({:enum, _})
false

  



  
    
      
    
    
      put_in_enum(enum, key, val)



        
          
        

    

  


  

Helper function to put a value into an enum, handling
not only maps and keyword lists but also structs.

  
    
  
  Examples


iex> Peri.put_in_enum(%{}, :hello, "world")
iex> Peri.put_in_enum(%{}, "hello", "world")
iex> Peri.put_in_enum(%User{}, :hello, "world")
iex> Peri.put_in_enum([], :hello, "world")

  



  
    
      
    
    
      to_changeset!(s, attrs)



        
          
        

    

  


  

      

          @spec to_changeset!(schema(), attrs :: map()) :: Ecto.Changeset.t()


      


Converts a Peri.schema() definition to an Ecto schemaless changesets.

  



  
    
      
    
    
      validate(schema, data)



        
          
        

    

  


  

Validates a given data map against a schema.
Returns {:ok, data} if the data is valid according to the schema, or {:error, errors} if there are validation errors.

  
    
  
  Parameters


	schema: The schema definition map.
	data: The data map to be validated.


  
    
  
  Examples


schema = %{
  name: :string,
  age: :integer,
  email: {:required, :string}
}

data = %{name: "John", age: 30, email: "john@example.com"}
Peri.validate(schema, data)
# => {:ok, %{name: "John", age: 30, email: "john@example.com"}}

invalid_data = %{name: "John", age: 30}
Peri.validate(schema, invalid_data)
# => {:error, [email: "is required"]}

  



  
    
      
    
    
      validate_schema(schema)



        
          
        

    

  


  

Validates a schema definition to ensure it adheres to the expected structure and types.
This function can handle both simple and complex schema definitions, including nested schemas, custom validation functions, and various type constraints.

  
    
  
  Parameters


	schema - The schema definition to be validated. It can be a map or a keyword list representing the schema.


  
    
  
  Returns


	{:ok, schema} - If the schema is valid, returns the original schema.
	{:error, errors} - If the schema is invalid, returns an error tuple with detailed error information.


  
    
  
  Examples


  Validating a simple schema:
  schema = %{
    name: :string,
    age: :integer,
    email: {:required, :string}
  }
  assert {:ok, ^schema} = validate_schema(schema)
  Validating a nested schema:
  schema = %{
    user: %{
      name: :string,
      profile: %{
        age: {:required, :integer},
        email: {:required, :string}
      }
    }
  }
  assert {:ok, ^schema} = validate_schema(schema)
  Handling invalid schema definition:
  schema = %{
    name: :str,
    age: :integer,
    email: {:required, :string}
  }
  assert {:error, _errors} = validate_schema(schema)

  


        

      


  

    
Peri.Generatable 
    



      
A module for generating sample data based on Peri schemas using StreamData.
This module provides functions to generate various types of data, conforming to the schema definitions given in Peri. It leverages the StreamData library to create streams of random data that match the specified types and constraints.

  
    
  
  Examples


iex> schema = %{
...>   name: :string,
...>   age: {:integer, {:gte, 18}},
...>   active: :boolean
...> }
iex> Peri.Generatable.gen(schema)
%StreamData{
  type: :fixed_map,
  data: %{name: StreamData.string(:alphanumeric), age: StreamData.filter(StreamData.integer(), &(&1 >= 18)), active: StreamData.boolean()}
}

      


      
        Summary


  
    Functions
  


    
      
        gen(schema)

      


        Generates a stream of data based on the given schema type.



    





      


      
        Functions

        


  
    
      
    
    
      gen(schema)



        
          
        

    

  


  

Generates a stream of data based on the given schema type.
This function provides various clauses to handle different types and constraints defined in Peri schemas. It uses StreamData to generate streams of random data conforming to the specified types and constraints.

  
    
  
  Parameters


	schema: The schema type to generate data for. It can be a simple type like :integer, :string, etc., or a complex type with constraints like {:integer, {:gte, 18}}.


  
    
  
  Returns


	A StreamData generator stream for the specified schema type.


  
    
  
  Examples


iex> Peri.Generatable.gen(:atom)
%StreamData{type: :atom, data: ...}

iex> Peri.Generatable.gen(:string)
%StreamData{type: :string, data: ...}

iex> Peri.Generatable.gen(:integer)
%StreamData{type: :integer, data: ...}

iex> Peri.Generatable.gen({:enum, [:admin, :user, :guest]})
%StreamData{type: :one_of, data: ...}

iex> Peri.Generatable.gen({:list, :integer})
%StreamData{type: :list_of, data: ...}

iex> Peri.Generatable.gen({:tuple, [:string, :integer]})
%StreamData{type: :tuple, data: ...}

iex> Peri.Generatable.gen({:integer, {:gt, 10}})
%StreamData{type: :filter, data: ...}

iex> Peri.Generatable.gen({:string, {:regex, ~r/^[a-z]+$/}})
%StreamData{type: :filter, data: ...}

iex> Peri.Generatable.gen({:either, {:integer, :string}})
%StreamData{type: :one_of, data: ...}

iex> Peri.Generatable.gen({:custom, {MyModule, :my_fun}})
%StreamData{type: :filter, data: ...}

iex> schema = %{name: :string, age: {:integer, {:gte, 18}}}
iex> Peri.Generatable.gen(schema)
%StreamData{type: :fixed_map, data: ...}

  


        

      


  

    
Peri.Parser 
    



      
The Peri.Parser module is responsible for managing the state during schema validation. 
It centralizes functions to handle updating data, adding errors, and managing the path within 
the data structure being validated.

  
    
  
  Struct


The Peri.Parser struct has the following fields:
	:data - The current state of the data being validated.
	:errors - A list of errors encountered during validation.
	:path - The current path within the data structure being validated.


      


      
        Summary


  
    Functions
  


    
      
        add_error(state, err)

      


        Adds an error to the parser state's list of errors.



    


    
      
        new(data, list)

      


        Initializes a new Peri.Parser struct with the given data.



    


    
      
        update_data(state, key, val)

      


        Updates the data in the parser state at the given key with the specified value.



    





      


      
        Functions

        


  
    
      
    
    
      add_error(state, err)



        
          
        

    

  


  

Adds an error to the parser state's list of errors.

  
    
  
  Parameters


	state - The current Peri.Parser state.
	err - The %Peri.Error{} struct representing the error to add.


  
    
  
  Examples


iex> state = Peri.Parser.new(%{name: "Alice"})
iex> error = %Peri.Error{path: [:name], message: "is required", content: []}
iex> Peri.Parser.add_error(state, error)
%Peri.Parser{data: %{name: "Alice"}, errors: [%Peri.Error{path: [:name], message: "is required", content: []}], path: []}

  



  
    
      
    
    
      new(data, list)



        
          
        

    

  


  

Initializes a new Peri.Parser struct with the given data.

  
    
  
  Parameters


	data - The initial data to be validated.


  
    
  
  Examples


iex> Peri.Parser.new(%{name: "Alice"})
%Peri.Parser{data: %{name: "Alice"}, errors: [], path: []}

  



  
    
      
    
    
      update_data(state, key, val)



        
          
        

    

  


  

Updates the data in the parser state at the given key with the specified value.

  
    
  
  Parameters


	state - The current Peri.Parser state.
	key - The key to update in the data.
	val - The value to set at the specified key.


  
    
  
  Examples


iex> state = Peri.Parser.new(%{name: "Alice"})
iex> Peri.Parser.update_data(state, :age, 30)
%Peri.Parser{data: %{name: "Alice", age: 30}, errors: [], path: []}

  


        

      


  

    
Peri.Error exception
    



      
Defines the structure and functions for handling validation errors in the Peri schema validation library.
The Peri.Error module encapsulates information about validation errors that occur during schema validation. Each error contains details about the path to the invalid data, the type of error, and any nested errors for complex or deeply nested schemas.

  
    
  
  Attributes


	:message - A human-readable message describing the error.
	:content - Additional information about the error, such as expected and actual values.
	:path - A list representing the path to the invalid data within the structure being validated.
	:key - The specific key or field that caused the error.
	:errors - A list of nested Peri.Error structs for detailed information about nested validation errors.


  
    
  
  Example


iex> error = %Peri.Error{
...>   message: "Validation failed",
...>   content: %{expected: :string, actual: :integer},
...>   path: [:user, :age],
...>   key: :age,
...>   errors: [
...>     %Peri.Error{
...>       message: "Expected type string, got integer",
...>       content: nil,
...>       path: [:user, :age],
...>       key: :age,
...>       errors: nil
...>     }
...>   ]
...> }
%Peri.Error{
  message: "Validation failed",
  content: %{expected: :string, actual: :integer},
  path: [:user, :age],
  key: :age,
  errors: [
    %Peri.Error{
      message: "Expected type string, got integer",
      content: nil,
      path: [:user, :age],
      key: :age,
      errors: nil
    }
  ]
}

  
    
  
  Functions


	error_to_map/1 - Converts a Peri.Error struct to a map, including nested errors.


      


      
        Summary


  
    Types
  


    
      
        t()

      


    





  
    Functions
  


    
      
        error_to_json(err)

      


    


    
      
        error_to_map(err)

      


        Recursively converts a Peri.Error struct into a map.



    


    
      
        format_error_message(reason, context)

      


    


    
      
        new_child(path, key, message, context)

      


        Creates a new child error with a path, key, message, and context.



    


    
      
        new_parent(path, key, errors)

      


        Creates a new parent error with nested errors.



    


    
      
        new_single(message, context)

      


        Creates a new single error with a formatted message and context.



    


    
      
        update_error_paths(error, new_path)

      


    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Peri.Error{
  __exception__: true,
  content: keyword(),
  errors: [t()] | nil,
  key: atom() | nil,
  message: String.t(),
  path: [atom()]
}


      



  


        

      

      
        Functions

        


  
    
      
    
    
      error_to_json(err)



        
          
        

    

  


  


  



  
    
      
    
    
      error_to_map(err)



        
          
        

    

  


  

Recursively converts a Peri.Error struct into a map.

  
    
  
  Parameters


	error - A Peri.Error struct to be transformed.


  
    
  
  Examples


iex> error = %Peri.Error{
...>   message: "Validation failed",
...>   content: %{expected: :string, actual: :integer},
...>   path: [:user, :age],
...>   key: :age,
...>   errors: [
...>     %Peri.Error{
...>       message: "Expected type string, got integer",
...>       content: nil,
...>       path: [:user, :age],
...>       key: :age,
...>       errors: nil
...>     }
...>   ]
...> }
iex> Peri.Error.error_to_map(error)
%{
  message: "Validation failed",
  content: %{expected: :string, actual: :integer},
  path: [:user, :age],
  key: :age,
  errors: [
    %{
      message: "Expected type string, got integer",
      content: nil,
      path: [:user, :age],
      key: :age,
      errors: nil
    }
  ]
}

  



  
    
      
    
    
      format_error_message(reason, context)



        
          
        

    

  


  


  



  
    
      
    
    
      new_child(path, key, message, context)



        
          
        

    

  


  

Creates a new child error with a path, key, message, and context.

  
    
  
  Parameters


	path - A list representing the path to the invalid data.
	key - The specific key or field that caused the error.
	message - A string template for the error message.
	context - A list of key-value pairs to replace in the message template.


  
    
  
  Examples


iex> Peri.Error.new_child([:user], :age, "Invalid value for %{field}", [field: "age"])
%Peri.Error{
  path: [:user, :age],
  key: :age,
  message: "Invalid value for age",
  content: %{field: "age"}
}

  



  
    
      
    
    
      new_parent(path, key, errors)



        
          
        

    

  


  

Creates a new parent error with nested errors.

  
    
  
  Parameters


	path - A list representing the path to the invalid data.
	key - The specific key or field that caused the error.
	errors - A list of nested Peri.Error structs.


  
    
  
  Examples


iex> Peri.Error.new_parent([:user], :age, [%Peri.Error{message: "Invalid age"}])
%Peri.Error{
  path: [:user, :age],
  key: :age,
  errors: [%Peri.Error{message: "Invalid age"}]
}

  



  
    
      
    
    
      new_single(message, context)



        
          
        

    

  


  

Creates a new single error with a formatted message and context.

  
    
  
  Parameters


	message - A string template for the error message.
	context - A list of key-value pairs to replace in the message template.


  
    
  
  Examples


iex> Peri.Error.new_single("Invalid value for %{field}", [field: "age"])
%Peri.Error{
  message: "Invalid value for age",
  content: %{field: "age"}
}

  



  
    
      
    
    
      update_error_paths(error, new_path)



        
          
        

    

  


  


  


        

      


  

    
Peri.InvalidSchema exception
    



      
Exception raised when an invalid schema is encountered.
This exception is raised with a list of Peri.Error structs,
providing a readable message overview of the validation errors.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %Peri.InvalidSchema{
  __exception__: true,
  errors: [Peri.Error.t()],
  message: String.t()
}

